/*
Samara Klar	
Proposal title: Gender Versus Party? Do Abortion Frames Affect Issue Engagement?


HYPOTHESES

Stated-Hyp1: "We propose to directly test whether framing abortion rights as a gender
issue or a partisan issue is more effective for mobilization and efficacy around abortion"...
"Our goal... is to consider the extent to which framing of abortion rights is more effective, meaning which frame is more likely to increase its importance and issue efficacy" (p.1)

	Test-Hyp1: Importance of abortion rights will be different in gender vs. partisan frame.
	Test-Hyp2: Efficacy regarding abortion rights will be different in gender vs. partisan frame.
	
Stated-Hyp2: "we expect that framing abortion as a partisan issue is likely to lead supporters of
abortion rights – both men and women – to perceive a greater capacity to affect change on the
issue." (p.2)
	// note: by "supporters of abortion rights", the author likely means liberals

	Test-Hyp3: Within liberals, efficacy will be higher in partisan vs. control frame.

Stated-Hyp3: "In contrast, we anticipate that the gender treatment may be especially demobilizing for
women who are supporters of abortion rights." (p.2)

	Test-Hyp4: Within liberals, the effect of gender frame on efficacy will be negatively moderated by gender (i.e. negative interaction effect of respondent gender and gender frame).


Stated-Hyp4: "We deliberately leave the effect of these treatments on those who oppose abortion rights as an open – and important -- research question" (p.2)

	Test-Hyp5: Within conservatives, importance of abortion rights will be different in partisan vs. control frame.
	Test-Hyp6: Within conservatives, importance of abortion rights will be different in gender vs. control frame.
	
	Test-Hyp7: Within conservatives, efficacy regarding abortion rights will be different in partisan vs. control frame.
	Test-Hyp8: Within conservatives, efficacy regarding abortion rights will be different in gender vs. control frame.	
	
********************************************************************************
NOTES

- the outcome (importance) and treatment are in the same question by design, 
as mentioned in the proposal.

- for stated hypothesis 2, reference group is not specified. We are assuming 
that it is the control group.

- for stated hypothesis 2-4, we are assuming that supporters and opposers of 
abortion are liberals and conservatives, respectively, as the proposal states: 
"we will rely on pre-measured ideology to capture abortion attitudes." (p. 2)

- there are a number of phantom outcomes. We are not pursuing them because 
they're not mentioned in the proposal.

- For hypothesis 4, seems like the authors want to test the differences for 
conservatives as well. We are comparing both treatment frames to the control, 
within conservatives.

*/

clear all
use "KlarS44.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

* which abortion frame shown

	lab def treat 0 "0 control" 1 "1 gender" 2 "2 partisan"
	gen treatment=0 if Q1_KLAR_CONTROL!=.
	replace treatment =1 if Q1_KLAR_GENDER!=.
	replace treatment=2 if Q1_KLAR_PARTISAN!=.
	lab val treatment treat
	tab treatment
	
* OUTCOME MEASURES

* importance
	clonevar importance = Q1_KLAR_CONTROL
	replace importance =Q1_KLAR_GENDER if treatment==1
	replace importance =Q1_KLAR_PARTISAN if treatment==2
	tab importance
	replace importance=. if importance >5

* efficacy
	tab Q2_KLAR
	clonevar efficacy = Q2_KLAR
 	replace efficacy = . if efficacy>7
	
* MODERATORS

* political ideology
	tab P_IDEO
	recode P_IDEO (1/3=1) (5/7=0) (*=.), gen(liberal)
	tab liberal
	
* gender
	tab GENDER
	recode GENDER (1=0) (2=1), gen(woman)
	tab woman
	
********************************************************************************

* ANALYSIS
	
*Test-Hyp1: Importance of abortion rights will be different in gender vs. partisan frame.
	reg importance i.treatment if treatment!=0
	// reject. p= 0.351
	tess 2.treatment, init(KlarS44) bonf(2)
	
*Test-Hyp2: Efficacy regarding abortion rights will be different in gender vs. partisan frame.
	reg efficacy i.treatment if treatment!=0
	// reject. p= 0.138
	tess 2.treatment, bonf(2)
	
* Test-Hyp3: Within liberals, efficacy will be higher in partisan vs. control frame.
	reg efficacy i.treatment if treatment!=1 & liberal==1
	// reject. p=0.617
	tess 2.treatment +, bonf(1)	

*Test-Hyp4: Within liberals, the effect of gender frame on efficacy will be negatively moderated by gender (i.e. negative interaction effect of respondent gender and gender frame).
	reg efficacy i.treatment##i.woman if treatment!=2 & liberal==1
	// reject. p=0.427
	tess 1.treatment#1.woman -, bonf(1)	
	
*Test-Hyp5: Within conservatives, importance of abortion rights will be different in partisan vs. control frame.
	reg importance i.treatment if treatment!=1 & liberal==0
	// reject. p=0.748
	tess 2.treatment, bonf(2)	
	
*Test-Hyp6: Within conservatives, importance of abortion rights will be different in gender vs. control frame.
	reg importance i.treatment if treatment!=2 & liberal==0
	// reject. p=0.462
	tess 1.treatment, bonf(2)
	
*Test-Hyp7: Within conservatives, efficacy regarding abortion rights will be different in partisan vs. control frame.
	reg efficacy i.treatment if treatment!=1 & liberal==0
	// reject. p=0.256
	tess 2.treatment, bonf(2)
	
*Test-Hyp8: Within conservatives, efficacy regarding abortion rights will be different in gender vs. control frame.	
	reg efficacy i.treatment if treatment!=2 & liberal==0
	// reject. p=0.911
	tess 1.treatment, bonf(2)	
